[Android Q&A, Help & Troubleshooting] :
Most ADB-FASTBOOT installers are provided in UPXed .EXE-file format hence it's not transparent what they are internally doing unless you decompile it.

So I decided to write the whole thing as a Windows CMD script thus everyone should can read and understand the coding:

@echo off & setlocal ENABLEDELAYEDEXPANSION
title Latest-ADB-FASTBOOT-Drivers-Installer
set "params=%*"
cd /d "%~dp0" && ( if exist "%TEMP%\getadmin.vbs" del /f /q "%TEMP%\getadmin.vbs" ) && fsutil dirty query %SYSTEMDRIVE% 1>nul 2>nul || ( echo Set UAC = CreateObject^("Shell.Application"^) : UAC.ShellExecute "cmd.exe", "/k cd ""%~sdp0"" && %~s0 %params%", "", "runas", 1 >> "%TEMP%\getadmin.vbs" && "%TEMP%\getadmin.vbs" && exit /B )
echo Downloading ...
%CD%\wget-x86 -q "https://dl.google.com/android/repository/platform-tools-latest-windows.zip" -O %TEMP%\platform-tools-latest-windows.zip
if NOT exist "%TEMP%\platform-tools-latest-windows.zip" ( echo FAILED & goto :end )
mkdir "%TEMP%\platform-tools-latest-windows" 2>nul >nul
echo Extracting ...
%CD%\7za-x86 x "%TEMP%\platform-tools-latest-windows.zip" -o"%TEMP%\platform-tools-latest-windows" >nul
set "src_path=%TEMP%\platform-tools-latest-windows\platform-tools"
if NOT exist "!src_path!\" ( echo FAILED & goto :end )
if NOT exist "!src_path!\adb.exe" ( echo FAILED & goto :end )
if NOT exist "!src_path!\AdbWinApi.dll" ( echo FAILED & goto :end )
if NOT exist "!src_path!\AdbWinUsbApi.dll" ( echo FAILED & goto :end )
if NOT exist "!src_path!\fastboot.exe" ( echo FAILED & goto :end )
echo Copying ...
set "tgt_path=%WINDIR%\ADB-FASTBOOT"
if NOT exist "!tgt_path!\" ( mkdir "!tgt_path!" )
copy /Y "!src_path!\adb.exe" "!tgt_path!\adb.exe" >nul
if NOT exist "!tgt_path!\adb.exe" ( echo FAILED & goto :end )
copy /Y "!src_path!\AdbWinApi.dll" "!tgt_path!\AdbWinApi.dll" >nul
if NOT exist "!tgt_path!\AdbWinApi.dll" ( echo FAILED & goto :end )
copy /Y "!src_path!\AdbWinUsbApi.dll" "!tgt_path!\AdbWinUsbApi.dll" >nul
if NOT exist "!tgt_path!\AdbWinUsbApi.dll" ( echo FAILED & goto :end )
copy /Y "!src_path!\fastboot.exe" "!tgt_path!\fastboot.exe" >nul
if NOT exist "!tgt_path!\fastboot.exe" ( echo FAILED & goto :end )
echo Adjusting PATH variable ...
echo %PATH% | findstr /C:"!tgt_path!" >nul
if !errorlevel! NEQ 0 ( setx /M PATH "%PATH%;!tgt_path!" )
:end
echo Housekeeping ...
if exist "%TEMP%\platform-tools-latest-windows.zip" ( del /F /Q "%TEMP%\platform-tools-latest-windows.zip" )
if exist "%TEMP%\platform-tools-latest-windows\" ( rmdir /S /Q "%TEMP%\platform-tools-latest-windows" )
echo Done
timeout /t 5 /nobreak >nul
endlocal & echo on
exit

Remark:

The big difference to others is that ADB & Fastboot drivers are updated in real time when this tool is used.

Important:

This tool doesn't install any Android USB driver because almost all OEMs have their own customized Android USB driver.

Requirement: Stable Internet connection.

Take note that with this script the mentioned drivers get installed in %WINDIR%\ADB-FASTBOOT what typically is C:\Windows\ADB-FASTBOOT hence they are reachable by everyone.

If you are interested in this script and in the required necessary executables then I let you know that all got packed into a .ZIP-file what is attached, wich you simply download and extract it to any location of your Windows computer, then in extracted folder run the Windows CMD housed therein. If installation was successful, please restart the Windows computer for changes to take effect.

.